TEXAS STATE UNIVERSITY SAN MARCOS

Department of Computer Science
San Marcos, TX 78666

Report Number TXSTATE-CS-TR-2006-3

Modeling Contract Bridge Bid Opening Strategies using the aspps System

Deborah East

2006-06-30

Modeling Contract Bridge Bid Opening Strategies using the *aspps* System

Deborah East
Department of Computer Science
Texas State University–San Marcos,
San Marcos TX 78666, USA

Abstract

Contract bridge is a popular and complex card game which has two distinct phases: the auction (bidding) and play. The result of the bidding is a contract (the number of tricks to be taken during play and the trump suit). Success depends as much on the contract as on the play. The difficulty in bidding has resulted in the development of strategies for bidding. In this paper, we model the opening strategies based on variations of precision and Standard American.

The language of logic PS^+ is used for encoding data and defining the rules for the bridge opening program. We take advantage of the nonmonotonic properties of the language of logic PS^+ to model contract bridge openings. The *aspps* system consists of two modules: *psgrnd* (for grounding a problem definition with the data of an instance of the problem) and the *aspps* solver. We show that the *aspps* system generates answer–sets for our bridge opening program where each answer–set is an opening bid.

1 Introduction

Contract bridge is a popular and complex card game which has two distinct phases: the auction (bidding) and play. The result of the bidding is a contract (the number of tricks to be taken during play and the trump suit). Success depends as much on the contract as on the play. A contract is reached by cooperation between partners where a dialog is limited to bids. During bidding each player can only see his own hand (13 cards) thus a player must draw inferences from his bids of his partner and to a lesser degree from the bids of his opponents. The difficulty in bidding has resulted in the development of strategies for bidding. In this paper, we model the opening strategies based on variations of precision [Wei, 1996] and Standard American [Downey and Pomer, 2004].

The language of logic PS^+ is used for encoding data and defining the rules for the bridge opening program. We take advantage of the nonmonotonic properties of the language of logic PS^+ when modeling the contract bridge opening program. The *aspps* system consists of two modules: *psgrnd* (for grounding a problem definition with the data of an instance of the problem) and the *aspps* solver. We show that the *aspps* system generates answer–sets for our bridge opening program where each answer–set is an opening bid.

We describe the language of logic PS^+ and the *aspps* system in Section 2. An introduction to contract bridge is then presented in Sections 3 followed by a discussion on bridge openings in Section 4. The modeling of the bridge opening program is presented in Section 5 followed by the results in Section 6. Last, conclusions and future work are discussed in Section 7.

2 The *aspps* system

The *aspps* system is an answer-set programming system based on the extended logic of propositional schemata (PS^+) [East and Truszczyński, 2001]. The language of logic PS^+ contains variable, constant and predicate symbols but not function symbols (theories are finite). Predicate symbols are separated into data predicates and program predicates. A problem instance is a set of ground atoms built of data predicates. A logic PS^+ program is a set of implications consisting of atoms constructed from both data and program predicates.

The language of logic PS^+ has built–in support of arithmetic and relational operators. Extensions to the logic of propositional schemata include *e–atoms* and *cardinality atoms*.

2.1 PS^+–theories

A theory in the logic PS^+ is a pair (D, P), where D is a set of ground atoms representing an *instance of a problem* (input data), and P is a set of PS^+-clauses representing a *program* (an abstraction of a problem). The meaning of a PS^+-theory $T = (D, P)$ is given by a *family* of PS^+-models [East and Truszczyński, 2001]. The key difference between the logic PS^+ and the logic of propositional schemata is in the definition of a model. Following the intuition that computation must not modify the data set, a set of ground atoms M is a model of a PS^+ theory (D, P) if M is a propositional model of the grounding of (D, P) and if it coincides with D on the part of the Herbrand Universe given by data predicates.

2.2 Computing models of PS^+–theories

The module, *psgrnd*[East *et al.*, 2004], instantiates a PS^+ theory with a data instance to produce grounded theories. The language of logic PS^+ accepted by *psgrnd* includes special constructs, such as those to model cardinality constraints on sets. The theories produced by *psgrnd* maintain the structure of these constructs. The grounded theories produced by *psgrnd* are input to *aspps* solver. The *aspps* solver accepts and takes advantage of special constructs in the ground theories to improve search. The *aspps* solver computes models of the ground theory and, hence, of the original theory, as well.

3 Bridge Introduction

Contract bridge (bridge) is a card game played with four people. A standard 52 card deck is used with each player dealt 13 cards (a hand). The players form two partnerships with partners sitting opposite each other (**North-South** or **East-West**). The game consists of two phases: an auction (bidding) and the play of the hand. This paper deals with the bidding phase, in particular, the opening bid.

During the auction each player can observe only their own hand and communication between partners is limited to bids. Each player bids in turn starting with the dealer. A bid can be *pass* or a bid level ($1, \ldots, 7$) and a suit (♣, ♢, ♡, ♠) or no trump ($\mathbf{NT}$), ranked low to high, respectively. Ordering of bids is as follows: 1 ♣ $<$ 1 ♢ $<$ 1 ♡ $<$ 1 ♠ $<$ $1\mathbf{NT}$ $<$ 2 ♣ $< \ldots <$ 7 $\mathbf{NT}$. As in any auction, each bid which is not *pass* must be higher than all previous bids. The contract is reached when a bid is followed by three *passes*. During play, the partnership winning the auction must take six tricks plus the number bid. i.e. A contract of 2 ♠ requires eight tricks to be taken or else a penalty is incurred.

We define here bridge terms which are used throughout this paper. The suits are divided into *minor* suits (♣ ♢) and *major* suits (♡ ♠). The evaluation of a bridge hands is based on the number of high card points (HCP) and distribution (the number of cards per suit). The cards are assigned points as follows: $ace = 4, king = 3, queen = 2, jack = 1$. A player *has* a suit if they have five cards in the suit. A *weak hand* typically has less than ten HCPs. A *long suit* is more than five cards in the suit. A *preempt* is a bid which is designed to keep opponents from reaching an optimal contract and is made, typically, with a *weak hand* and *long suit*. A hand with at least 2 cards and no more than 4 cards in each suit is *balanced*. A *convention* or *conventional bid* is a bid which has a meaning agreed upon by a partnership which is not the usually meaning of the bid.

4 Bridge Openings

The evaluation of a bridge hands is based on the number of high card points (HCP) and the distribution (the number of cards per suit). The HCP count along with the distribution of cards in each suit determine whether to open a hand and if the hand is opened which suit is bid. In general, a hand with 12 or more HCPs is considered an opening hand while a hand with fewer points may be opened if there is a *long suit* or if the bid is a *convention*.

Several factors affect bidding strategies. The major suits score more points per trick than the minors suits. Thus major suit contracts are preferred over minor. Trick for $\mathbf{NT}$ score higher than for major suits. However, $\mathbf{NT}$ may or may not be preferred. Since there is not a trump suit, some players consider a $\mathbf{NT}$ contract more difficult to play. The partnership which wins the auction has the advantage of playing in their suit, but, penalties are enacted if the contract is not made. Thus, the goal is to win the contract at a level where one can reasonably expect to take the required number of tricks.

There are numerous general approaches and variations for bidding strategies. We have chosen two general approaches to model in this paper: precision and Standard American (SA). The SA is a natural approach which minimizes the use of *conventional bids*. As explained in the previous sections, conventional bids are those which have a meaning agreed upon by the partnership. A hand with at least 12 HCPs is considered an opening hand (at the one level) in SA with a major suit bid requiring at least five cards in the major suit and a minor suit opening requiring at least three cards in the minor suit. An opening one $\mathbf{NT}$ range is 15 to 17 HCPs and typically should not have five cards in a major suit. The range for two and three level $\mathbf{NT}$ bids are 20 to 21 and 26 to 27 respectively. An opening 2 ♣ is the main conventional bid in SA. It implies a strong hand (22 or more HCPs) which does not have to have a minimum number of clubs. The 2 ♢, 2 ♡ and 2 ♠ bids are weak (6 to 11 HCPs) with at least 6 cards in the bid suit. Hands with fewer HCPs and at least 7 cards in a suit can be opened at the three level. i.e. 3 ♣, 3 ♢, 3 ♡ or 3 ♠

The precision approach tries to give more information within the restricted dialog of bids. The opening one level majors bids are limited to at most 15 HCPs and require five cards in the bid major. The opening bid of 1 ♣ has a range of 16 or more HCPs and implies nothing about the distribution. The opening bid of 1 ♢ can be used for a hand with a five card diamond suit or a *balanced hand* with 15 to 17 HCPs. A 1 $\mathbf{NT}$ can be opened with 12 to 14 HCP. In this variation of precision a 2 ♣ opening reflects a range of 12 to 15 HCPs and at least 6 clubs. The other two level bids, 2 ♢, 2 ♡ and 2 ♠, indicate *weak hands* with a range of 8 to 11 HCPs, at least six cards in the bid suit and a maximum of two cards in any unbid major suit. Hands with fewer HCPs and at least 7 cards in the bid suit can be opened at the three level as in SA.

5 Bridge Opening Program

There are two key issues addressed in the bridge opening program. First is the abstraction of the strategy from the program. Strategies for deciding opening bids are generally based on high card point (HCP) count and suit lengths. The program does not use HCP values or suit length values in its rules. The specific values are given as data. Thus the program is independent of any particular bidding strategy. Second, the nonmonotonicity of the language of logic PS^+ allows us to model rules which eliminate bids that do not conform to our strategy instead of modeling rules which conforms to a strategy.

5.1 Game Data

The *game data* is modeled to be independent of a particular strategy or hand. Unary predicates are used to define *game* data. We define for each suit whether it is a major or minor suit:

```
major(spade). major(heart).
minor(club). minor(diamond).
```

Because bids include **NT** as well as suits, we define bid suits to include **NT**:

```
suits(club). suits(diamond).
suits(heart). suits(spade).
suits(notrump).
```

We define the bid levels and the ranges for HCPs and suit lengths:

```
level(1..7). point(0..40). length(1..13).
```

5.2 Strategy data

The strategy is given as data which specifies ranges of HCP, bid levels and minimum lengths. The **NT** opening data predicate $notrumpopen$ is a triple (V, M, N) where V is the bid level, M is the minimum HCP and N is the maximum HCP for the bid. In addition, a unary data predicate $notrumpmin$ represents the minimum number of cards required for each suit. Most strategies prefer major suit contracts when possible thus we set a maximum major card length using the unary data predicate $ntmaxmajor$. The data predicate *open* used for opening suits is a $5 - tuple$, (V, S, M, N, K) where V is the bid level, S is the bid suit, M is the minimum HCP, N is the maximum HCP and K is the minimum number of cards in suit S. The suits and **NT** are handled separately since K is not applicable to **NT** bids.

Partner agreements may include preferences for a major if both are equal in length. The binary data $majorprefer(S, V)$ indicates a preference for suit S at bid level V.

We want to eliminate opening at levels higher than specified preempts. The unary data predicate *preempt* represents the maximum opening bid and reflects the highest specified bid.

The data predicate $convention$ is a pair (S, V) where S is a suit and V is a bid level. We include handling of conventions here so that *conventional bids* are not eliminated.

Standard American Data

The data values presented here for SA are based on ACBL's Standard American Yellow Card (www.acbl.org) with some variation.

```
notrumpopen(1,15,17).
notrumpopen(2,20,21).
notrumpopen(3,26,27).
notrumpmin(2).
ntmaxmajor(4).
open(1,spade,12,21,5).
open(1,heart,12,21,5).
open(1,diamond,12,21,3).
open(1,club,12,21,3).
open(2,spade,6,11,6).
open(2,heart,6,11,6).
open(2,diamond,6,11,6).
open(2,club,22,37,0).
open(3,spade,3,11,7).
open(3,heart,3,11,7).
open(3,diamond,3,11,7).
open(3,club,8,11,6).
majorprefer(heart,1).
convention(club,2).
preempt(3).
```

Precision Data

The goal of the precision approach is to provide more information to partners at lower bidding levels. The range of HCPs for major suit at the one level are much narrower than for SA, the strong hand is represented at the one level, 1 ♣, instead of at the two level. A 12 to 14 HCP for 1 **NT** and a 15 to 17 HCP **NT** hand being bid as 1 ♢. An 11 to 14 HCP hand with a diamond suit is also bid as 1 ♢.

```
notrumpopen(1,12,14).
notrumpopen(2,21,22).
notrumpopen(3,25,27).
notrumpopen(4,27,29).
notrumpmin(1).
ntmaxmajor(5).
open(1,spade,12,17,5).
open(1,heart,12,17,5).
open(1,diamond,11,17,2).
open(1,club,18,37,0).
open(2,spade,8,11,6).
open(2,heart,8,11,6).
open(2,diamond,8,11,6).
open(2,club,11,15,6).
open(3,spade,3,11,7).
open(3,heart,3,11,7).
open(3,diamond,3,11,7).
open(3,club,3,11,7).
open(4,spade,11,15,8).
open(4,heart,11,15,8).
open(4,diamond,11,15,8).
open(4,club,11,15,8).
convention(club,1).
majorprefer(spade,1).
preempt(4).
```

5.3 Hand Data

Hand data defines the total high card points, the distribution of high card points and the card length per suit. The data predicate *ttlhcp* is a unary predicate and is used to represent the total high card points when evaluated as $ace = 4, king = 3, queen = 2, jack = 1$. Each suit in the hand is represented by a triple (X, M, J) where X is the suit, M is the HCPs in the suit and J is the length of the suit.

The following examples have a total HCP of 14 which according to general guidelines should be opened. Different strategies result in different opening bids. For example, this hand would be opened 1 ♣ using data for SA and 1 **NT** using data for precision.

```
ttlhcp(14).
```

```
suit(club,4,4).
suit(diamond,2,2).
suit(heart,1,3).
suit(spade,4,4).
```

5.4 Bridge Program Rules

The preamble in the rule file defines the program predicates bid and pass and declares variables using data predicates. The variables are used in the program rules.

```
pred bid(suits,level).
pred pass.

var points P,Q,R,S.
var length J,K,L.
var suits X,Y.
var level U,V.
```

1. This rule requires that there is exactly one bid or pass.

```
1{pass;bid(_,_)}1.
```

2. This rule prevents a **NT** bid if the hand has any suit with card length less than minimum specified by the strategy file unless the **NT** bid is a convention.

```
   suit(X,P,K),
   notrumpmin(J),
   K < J,
   bid(notrump,V) ->.
convention(notrump,V).
```

3. This rule prevents a **NT** bid if the hand has a major suit with length greater than the maximum major length specified by the strategy file unless the **NT** bid is a convention.

```
   suit(X,P,K),
   ntmaxmajor(J),
   major(X),
   K > J,
   bid(notrump,V) ->.
convention(notrump,V).
```

4. This rule eliminates **NT** bids at level V if the hand has less high card points than the minimum required for a **NT** bid at level V.

```
   notrumpopen(V,P,Q),
   ttlhcp(R),
   R<P,
   bid(notrump,V) ->.
```

5. This rule eliminates **NT** bids at level V if the hand has more high card points than the maximum for a **NT** bid at level V.

```
   notrumpopen(V,P,Q),
   ttlhcp(R),
   R>Q,
   bid(notrump,V) ->.
```

6. This rule prevents a hand from passing if the hand has more high card points than the minimum for a **NT** bid at any level.

```
   notrumpopen(V,P,Q),
   ttlhcp(R),
   R>P,
   pass ->.
```

7. This rule eliminates bids of suit X at level U if the hand has too few HCP.

```
   open(U,X,P,Q,J),
   ttlhcp(R),
   R < P,
   bid(X,U) ->.
```

8. This rule eliminates bids of suit X at level U if the hand has too many HCP.

```
   open(U,X,P,Q,J),
   ttlhcp(R),
   R > Q,
   bid(X,U) ->.
```

9. This rule eliminates bids of suit X at level U if suit X has too few cards in suit X.

```
   open(U,X,P,Q,J),
   suit(X,R,K),
   K < J,
   bid(X,U) ->.
```

10. This rule prevents passing if the hand contains opening HCP count and an opening suit.

```
   open(U,X,P,Q,J),
   ttlhcp(R),
   R >= P,
   suit(X,S,K),
   K >= J,
   pass ->.
```

11. This rule allows a strategy to prefer one major suit over the other.

```
   suit(X,P,K),
   open(U,X,R,S,L),
   K>=L,
   ttlhcp(Q),
   Q>=R,
   major(X),
   major(Y),
   X != Y,
   bid(Y,U) ->
majorprefer(Y,U).
```

12. This bid eliminates a minor suit bid if a major suit has sufficient length to open unless the minor suit bid is a convention.

```
   suit(X,P,K),
   open(U,X,R,S,L),
   ttlhcp(Q),
   L<=K,
   Q>=R,
   Q<=S,
   major(X),
   minor(Y),
```

```
    bid(Y,U) ->
convention(Y,U).
```

13. This rule eliminates all bids higher than the specified preempt level. Without this rule we would have to specify all the higher level bids in the strategy data even if our strategy does not allow them.

```
    bid(X,U),
    U > V,
    preempt(V) ->.
```

The rules for the bridge opening program eliminate bids which are inconsistent with either the general approach given by the strategy data or with a particular convention specified by the strategy data.

6 Results

We show the result from a set of 12 hand–crafted hands. The first hand results in **pass** for both SA and precision. The remaining hands require an opening bid and the *aspps* system finds answer–sets for both strategies with each hand.

It is typical for the grounder to find a solution since the rules are designed to force to false atoms which are contradicted by the instantiation of data and program predicates. In the case where the grounder does not find a solution and there are no inconsistencies then there are multiple answer-sets. Any inconsistency found by the grounder or solver is a result of incorrect data for the strategy since there must always be at least one answer-set.

```
ttlhcp(10).
suit(club,3,3).
suit(diamond,3,2).
suit(heart,2,5).
suit(spade,2,3).
```

Standard: pass
Precision: pass
Total HCPs is below minimum opening without a long suit which would allow a preemptive bid.

```
ttlhcp(21).
suit(club,7,4).
suit(diamond,5,3).
suit(heart,5,3).
suit(spade,4,3).
```

Standard: 1 ♣, 1 ♢, 2 $\mathbf{NT}$
Precision: 1 ♣, 2 $\mathbf{NT}$
The HCP is below the minimum for the conventional 2 ♣ bid in SA but is above the HCP for the precision conventional 1 ♣ bid.

```
ttlhcp(22).
suit(club,8,3).
suit(diamond,5,2).
suit(heart,5,5).
suit(spade,4,3).
```

Standard: 2 ♣
Precision: 1 ♣, 2 $\mathbf{NT}$
Both strategies have the minimum HCP met for their conventional bids.

```
ttlhcp(14).
suit(club,4,4).
suit(diamond,2,2).
suit(heart,1,3).
suit(spade,4,4).
```

Standard: 1 ♣
Precision: 1 ♢, 1 $\mathbf{NT}$
HCP minimum is not met for SA 1 $\mathbf{NT}$ but it is met for precision. Precision would also not eliminate a 1 ♢ bid.

```
ttlhcp(16).
suit(club,4,4).
suit(diamond,4,2).
suit(heart,4,3).
suit(spade,4,4).
```

Standard: 1 ♣, 1 $\mathbf{NT}$
Precision: 1 ♢
SA does not eliminate either 1 ♣ or 1 $\mathbf{NT}$ for this hand. Precision 1 $\mathbf{NT}$ is too weak and a rebid of NT after the initial 1 ♢ bid allows an extra NT point range.

```
ttlhcp(18).
suit(club,5,3).
suit(diamond,3,3).
suit(heart,3,2).
suit(spade,7,5).
```

Standard: 1 ♠
Precision: 1 ♣
The HCPs here is within the 1 level for SAs major bids but meets the minimum HCPs for precisions conventional 1 ♣ bid.

```
ttlhcp(15).
suit(club,3,2).
suit(diamond,3,1).
suit(heart,3,5).
suit(spade,7,5).
```

Standard: 1 ♡
Precision: 1 ♠
Illustrations major preferences in the strategies.

```
ttlhcp(22).
suit(club,8,4).
suit(diamond,5,3).
suit(heart,5,3).
suit(spade,4,3).
```

Standard: 2 ♣
Precision: 1 ♣, 2 $\mathbf{NT}$
Meets the minimum HCPs for both SA and precision conventional bids. It also falls with precisions 2 $\mathbf{NT}$ range.

```
ttlhcp(20).
suit(club,8,4).
suit(diamond,5,3).
suit(heart,4,3).
suit(spade,3,3).
```

Standard: 1 ♣, 1 ♢, 2 $\mathbf{NT}$
Precision: 1 ♣
SA has three bids which are valid for this hand while precisions only bid is the conventional 1 ♣ bid.

```
ttlhcp(18).
suit(club,7,2).
suit(diamond,5,5).
suit(heart,3,3).
suit(spade,3,3).
```

Standard: 1 $\diamondsuit$
Precision: 1 $\clubsuit$
SA has not met the minimum HCP for its conventional bid whereas precision has.

```
ttlhcp(14).
suit(club,4,5).
suit(diamond,4,2).
suit(heart,4,3).
suit(spade,2,3).
```

Standard: 1 $\clubsuit$
Precision: 1 $\diamondsuit$, 1 $\mathbf{NT}$
SA has not met the minimum HCP for its 1 $\mathbf{NT}$ bid whereas precision has.

```
ttlhcp(15).
suit(club,4,2).
suit(diamond,4,6).
suit(heart,4,3).
suit(spade,4,2).
```

Standard: 1 $\diamondsuit$, 1 $\mathbf{NT}$
Precision: 1 $\diamondsuit$
SA has met the minimum HCP for its 1 $\mathbf{NT}$ bid and precision must rebid NT after the 1 $\diamondsuit$ to show the NT at a higher HCP than its 1 $\mathbf{NT}$.

7 Conclusions and Future Work

In this paper, we show that by encoding the bidding strategy as data, the bridge opening program is independent of any particular strategy. We can change a strategy by modifying the data. Thus, distinct partnerships can use the program each employing their own strategies.

Future work includes modeling responses to opening bids, over-calls and rebids such that sequences of bids can be generated with each bid is a separate execution of the system. Each bid in the sequence is based upon players hand, previous partnership bids and opponent bids. In addition to the bridge opening program, *aspps* programs would be required for the cases where the bidder was originally the opener, over-caller or responder.

The goal is to generate a sequence of bids where any bid can be either from a human or from an answer-set resulting from the bridge bidding program modeled in the language of logic PS^+ and computed using the *aspps* system.

References

[Downey and Pomer, 2004] N. Downey and E. Pomer. *Standard Bidding with SAYC.* MasterPoint Press, 2004.

[East and Truszczyński, 2001] D. East and M. Truszczyński. Propositional satisfiability in answer set programming. In *Proccedings of KI-2001).* Springer Verlag, September 2001.

[East *et al.*, 2004] D. East, M. Iakhiaev, A. Mikitiuk, and M. Truszczyński. Tools for modeling and solving search problems. In *Third International Workshop on Modelling and Reformulating Constraint Satisfaction Problems*, Toronto, 2004.

[Wei, 1996] C. C. Wei. *Precision Bidding System in Bridge*. Baron/Barclay Bridge Supplies, 1996.